
1、减少盘整行情中的交易次数-PANZHENG

很多趋势模型，在行情出现趋势的时候，都可以很好的抓住趋势，实现盈利，但长期运行下

来，最终的结果却是小赚甚至亏钱，问题出在哪里？

原因在于，盘整行情中模型在不断的反复交易，而盘整中的交易都是不盈利甚至亏损的，行

情中绝大部分又都是盘整行情，长时间的连续小亏损导致之前的利润全部回吐

图 1

图 1 中的模型，在红圈位置出现的较大回撤，就是由震荡行情导致，放大的细节见图 2

图 2

如何避免在这样的震荡行情中多次交易，是我们优化策略的首要任务。

由于很多投资者没有很好的方法去过滤震荡行情，因此文华财经研发了 PANZHENG 函数，帮

助大家过滤震荡行情，函数说明如下：

PANZHENG 判断当前行情是否为盘整

用法：返回 1:表示盘整，返回 0:表示不是盘整。

注：

这个函数的目的是用于判断当根 k线是否盘整状态，是否适合做开仓操作，从而优化趋势模型，避免在盘

整阶段频繁交易。

PANZHENG 函数具体的编写方法和效果，我们看一下下面的案例：

作用一：增加盈利

一个普通的均线模型，加载在股指上的效果

MA1:=MA(C,5);

MA2:=MA(C,10);

CROSS(MA1,MA2),BPK;

CROSS(MA2,MA1),SPK;

AUTOFILTER;



3

这段行情中实现盈利 77040 元，见图 3

图 3

为了过滤掉模型开仓时候的盘整行情，我们把模型分成多空两个模型并分别加入 PANZHENG

函数，做多模型代码如下：

MA1:=MA(C,5);

MA2:=MA(C,10);

CROSS(MA1,MA2)&&PANZHENG=0,BK;

CROSS(MA2,MA1),SP;

AUTOFILTER;

注意红色部分是加入的盘整函数，过滤盘整行情中的开仓信号

做多实现盈利 179580 元，见图 4

图 4

做空模型代码如下：

MA1:=MA(C,5);

MA2:=MA(C,10);

CROSS(MA2,MA1)&&PANZHENG=0,SK;

CROSS(MA1,MA2),BP;

AUTOFILTER;

加入盘整函数后，做空亏损 44100 元，见图 5

图 5

未加入盘整函数前，这段行情中多空共实现盈利 77040 元

加入盘整函数后，做多实现盈利 179580 元，做空亏损 44100 元

这段行情中多空共实现盈利 135480 元

加入盘整函数后，盘整行情交易次数大量减少，从而减少了亏损

总盈利提升 76%

作用二：减小最大回撤

普通均线模型加载到 PTA 指数

代码如下：

MA10:=MA(C,10);

C>MA10,BPK;//价格大于 10 周期均线，做多

C<MA10,SPK;//价格小于 10 周期均线，做空

AUTOFILTER;

2010.1.1 至今的测试结果，模型虽然有年化 55%的收益率，但也有 65%的权益最大回撤，如

此大的回撤导致模型的实际可用性大大降低，见图 6

图 6

加入 PANZHENG 函数后，代码如下

MA10:=MA(C,10);

C>MA10&&PANZHENG=0,BPK;//非盘整行情中，价格大于 10 周期均线，做多

C<MA10 &&PANZHENG=0,SPK;//非盘整行情中，价格小于 10 周期均线，做空

AUTOFILTER;

加载到 PTA 指数，2010.1.1 至今的测试结果

胜率提升 14%

盈利率提升 37%

最大回撤减少 45%

年化盈利率提升 21%

单次交易盈利能力提升 40%

减少盘整行情中的交易次数后，不仅仅盈利能力得到提升，模型的稳定性同时也得到大幅度

提升，大大提高了模型的可执行性，加入 PANZHENG 函数后的测试报告见图 7

图 7

2、优化进出场点-CHECKSIG

大部分趋势模型采用趋势突破或者趋势跟踪的方法捕捉趋势，通常情况下，信号都出现在一

根较长的 k线上，如果使用收盘价模型，入场点在下根 k线的开盘价上，因此会错过突破这

根长 k线的最佳入场时间，无形中损失掉很大一笔到手的利润；出场时也是如此，那么怎么

才能以最优的价格进出场，得到更多的利润呢？

如何实现在 k 线没有走完之前进出场呢？

我们对比一下下面的两张图片图 8和图 9，同样一个模型，分别加载到同一段数据上的结果：

图 8 是收盘价模型，红色虚线为开仓价格位置。红圈中，由于 k 线当根涨幅较大或者第二根

跳空，损失较多利润甚至盈利变为亏损

图 9 是指令价模型，红色虚线为开仓价格位置。与图 8 中的开仓位置比较，图 9 的盈利能力

要大大提升，因为指令价模型在条件满足的瞬间即可进场交易，不必等到出现信号的 k 线走

完才交易，交易效率大大提升（绿色虚线代表的平仓位置也是同样效果）。

图 8

图 9

模型中加入 CHECKSIG 函数，就可以实现图 9中的出信号立即发出委托。函数说明如下（更

加具体函数说明请在函数列表中查阅）：

CHECKSIG 设置信号确认与复核的指令价方式（TICK 逐笔回测，可设置回测精度）

用法：

CHECKSIG(SIG,MODE1,TIME1,MODE2,TIME2,INTERVAL);SIG 为信号,MODE1 为信号确认方式,TIME1 信号确认

时间,MODE2 信号复核方式,TIME2 信号复核时间,INTERVAL 数据时间间隔。

通过调整 INTERVAL 参数，模型可设置不同数据快照频率进行回测。

CHECKSIG 函数的编写方法：

CHECKSIG(BK,'B',5,'C',0,3);//设置 BK 信号的信号执行方式为：K 线走完前 5*3=15 秒下

单，不进行复核，每隔 3秒计算一次信号。

CHECKSIG(SK,'B',5,'C',0,3);//设置 SK 信号的信号执行方式为：K 线走完前 5*3=15 秒下

单，不进行复核，每隔 3秒计算一次信号。

CHECKSIG(SP,'A',0,'C',5,3);//设置 SP 信号的信号执行方式为：出信号立即下单，下单后

5*3=15 秒进行复核，每隔 3 秒计算一次信号。

CHECKSIG(BP,'A',0,'C',5,3);//设置 BP 信号的信号执行方式为：出信号立即下单，下单后

5*3=15 秒进行复核，每隔 3 秒计算一次信号。

模型中加入 CHECKSIG，在实现指令价模型的同时，可以针对不同指令实现不同的信号复核

方式，让交易策略的实现更加灵活：

案例：如图 10，交易中开仓多采用 k 线走完前 N 秒下单，以保证进场准确性，避免因行情

波动而频繁开平仓。从而增加交易成本

平仓多采用出信号立即下单执行，保证单子能够及时出场，避免持仓过久导致无法及时止损

和降低盈利空间

图 10

图 10 模型源码如下：

MA5^^MA(C,5);

MA10^^MA(C,10);

RSV:=(CLOSE-LLV(LOW,9))/(HHV(HIGH,9)-LLV(LOW,9))*100;

K:=SMA(RSV,3,1);

D:=SMA(K,3,1);

EVERY(MA5>MA10,5)&&CROSSUP(K,D),BK;

EVERY(MA5<MA10,5)&&CROSSDOWN(K,D),SK;

BARSBK>2&&C<=BKHIGH-10*MINPRICE,SP;

BARSSK>2&&C>=SKLOW+10*MINPRICE,BP;

CHECKSIG(BK,'B',1,'C',0,0); //设置 BK 信号执行方式为：K 线走完前 1 秒下单，不进行

复核

CHECKSIG(SK,'B',1,'C',0,0); //设置 SK 信号执行方式为：K 线走完前 1 秒下单，不进行

复核

CHECKSIG(SP,'A',0,'C',0,0); //设置 SP 信号执行方式为：出信号立即下单，不进行复核

CHECKSIG(BP,'A',0,'C',0,0); //设置 BP 信号执行方式为：出信号立即下单，不进行复核

AUTOFILTER;

我们在追求模型更加的进出场点的同时，要清楚的认识到并不是所有的模型都是指令价优于

收盘价的。究竟用指令价效果好还是收盘价效果好，还要根据交易策略决定。一些交易逻辑

简单的模型，指令价或者收盘价效果区别较小。但收盘价模型无法处理更加细致的交易逻辑，

就需要采用指令价了。

图 11 中的模型信号消失的概率较小，信号消失成本较小，所以指令价优于收盘价。

图 12 中的模型则相反，有非常大的信号消失成本，所以还是收盘价要好一些

图 11

图 12

TICK 回测是最准确的回测方式：

图 13 中的模型，平仓条件为小于变量 LL，历史回测：

LL 数值为 2585.8 价格为 2585.6 时满足 SP 信号条件，从盘口 TICK 数据我们可以看到，盘

中价格断档，所以回测结果为 9:37:24 时的价格 2585.4，而非满足条件时的 2585.6，因为

盘中没有这个价格。

图 13

是否支持指令价委托并不重要，重要的是是否支持回测

WH8.2 是国内程序化平台中唯一提供指令价模型 tick 回测的程序化交易软件。是历史回测

最精准的程序化交易软件。

3、在当根 k 线上开始新一轮交易-MULTSIG

期货市场价格瞬息万变，经常会出现价格瞬间拉升，接着就很快反抽的情况。收盘价模型在

行情拉升时出现开仓信号，如果遇到反抽行情，即使能够及时平仓，也不能够马上开始下一

轮交易抓住反抽行情，因为收盘价模型还是执行上根 k线的交易指令，如何才能做到非反手

指令在当根 k 线上开始新一轮交易呢？

图 14 收盘价模型 图 15 指令价模型

上图的交易策略，在这种秒杀行情中，行情已经逆转下跌，收盘价模型（图 14），还在执

行上根 k 线的买开仓指令，显然是错误的，导致大幅度亏损。而指令价模型（图 15），则

可以当根 k线同时完成买开仓、平仓、反向卖开仓一系列交易，即保证既得利润，又及时抓

住了反向行情。

模型中加入 MULTSIG 函数，可以实现图 15 中的指令价模型，即在一根 k 线上交易多次。实

现更精致的交易策略，可以很好的规避掉秒杀行情。

MULTSIG 函数说明：

MULTSIG(Sec1,Sec2,N,INTERVAL) 设置一根 k线多信号的指令价方式（TICK 逐笔回测，可设置回测精度）

用法：

MULTSIG(Sec1,Sec2,N,INTERVAL) 设置信号复核确认方式，开仓信号出信号 Sec1 秒下单不复核,平仓信号

出信号 Sec2 秒下单不复核，一根 K线上最大的信号个数为 N,INTERVAL 为数据时间间隔。

通过调整 INTERVAL 参数，模型可设置不同数据快照频率进行回测 。

MULTSIG 函数的编写方法，可以参考下面的模型：

C>REF(H,1),BK;//价格大于上一根 k 线最高价，开多仓

C<BKPRICE-3*MINPRICE,SP;//亏损 3 点止损

MULTSIG(0,0,3,0);//开仓信号，出信号立即下单，不复核；平仓信号出信号立即下单，不

复核。一根 K 线上最大信号个数为 3。

AUTOFILTER;

案例：

图 16 是 IF 加权 15 分钟周期，2016.1.4-2016.1.14 这段时间的走势，这段时间出现多次急

涨急跌行情

图 16

收盘价模型，开仓后只能在下一根 k线上发出平仓信号进行止损，导致在这种行情中出现亏

损，测试报告见图 17

RSV:=(CLOSE-LLV(LOW,9))/(HHV(HIGH,9)-LLV(LOW,9))*100;

K:=SMA(RSV,3,1);

D:=SMA(K,3,1);

CROSS(K,D),SK;

CROSS(D,K),BK;

(O-L>3*(C-O)||(C-L>3*(O-C)))&&V>REF(V,1)*1.5,BP;

(O-L>3*(C-O)||(C-L>3*(O-C)))&&V>REF(V,1)*1.5&&COUNTSIG(BK,1)=0,BK;

(O-L>3*(C-O)||(C-L>3*(O-C)))&&V>REF(V,1)*1.5,SP;

(O-L>3*(C-O)||(C-L>3*(O-C)))&&V>REF(V,1)*1.5&&COUNTSIG(SK,1)=0,SK;

AUTOFILTER;

图 17

我们在模型中加入 MULTSIG，让模型变成指令价模型，并设置一根 k 线上可以出现 3 个信号，

就实现了在开仓 k 线上进行及时的平仓并且反向开仓，模型如下：

RSV:=(CLOSE-LLV(LOW,9))/(HHV(HIGH,9)-LLV(LOW,9))*100;

K:=SMA(RSV,3,1);

D:=SMA(K,3,1);

CROSS(K,D),SK;

CROSS(D,K),BK;

(O-L>3*(C-O)||(C-L>3*(O-C)))&&V>REF(V,1)*1.5,BP;

(O-L>3*(C-O)||(C-L>3*(O-C)))&&V>REF(V,1)*1.5&&COUNTSIG(BK,1)=0,BK;

(O-L>3*(C-O)||(C-L>3*(O-C)))&&V>REF(V,1)*1.5,SP;

(O-L>3*(C-O)||(C-L>3*(O-C)))&&V>REF(V,1)*1.5&&COUNTSIG(SK,1)=0,SK;

MULTSIG(0,0,3,10);

AUTOFILTER;

从测试报告我们可以看到，加入 MULTSIG 后，模型可以及时进行平仓并且反向抓住交易机会，

同样的这段时间内，亏损变成盈利，见图 18

图 18

4、解决中长线趋势交易换月跳空的问题-TRADE_OTHER

市场主流的交易理念中，趋势交易由于可容纳资金量较大，交易方法容易学习掌握，在市场

中备受青睐。而且绝大部分趋势交易系统都可以进行量化，进行程序化交易，所以很多程序

化交易者把趋势交易当作交易组合中一个必不可少的策略，甚至是主要策略！

在研发趋势交易系统的过程中，需要较长的历史数据进行模型的回测。那么在回测的时候，

选择具体月份合约、主连合约、指数合约中哪一个数据进行回测才能得到更真实的效果？

我们通过下面案例看一下：

下面是一个简单的均线趋势系统，代码如下：

MA1:MA(CLOSE,5);

MA2:MA(CLOSE,10);

MA3:MA(CLOSE,60);

C>MA3&&PANZHENG=0&&MA1>MA2,BK;

C<MA3&&PANZHENG=0&&MA1<MA2,SK;

C>MA1&&BARSSK>10,BP;

C<MA1&&BARSBK>10,SP;

C>HV(C,5)&&BARSSK>5,BP;

C<LV(C,5)&&BARSBK>5,SP;

AUTOFILTER;

首先，我们把这个趋势模型加载到具体月份的合约上，我们选择 IF1602 合约的 15 分钟 K

线，得到的测试结果如下,见图 19：

图 19

从上面的测试报告中我们可以看出，这个趋势系统加载到 IF1602 合约的 15 分钟 K 线图上得

出的结果还是不错的，那这个模型是否可以用于实盘呢？

我们都知道，对于趋势模型的回测，应该用更多的历史数据。而 IF1602 的历史数据只有

2015.12.21-2016.1.21 短短一个月的时间，而这段时间又是以单边下跌的趋势为主，并不

能说明模型在各种不同行情中的表现，所以上面的测试报告并不能真实的说明模型的盈利能

力。

为了测试更多的数据，我们把模型加载到主连合约上，选取的测试时间是

2015.1.5--2016.1.21，得到图 20 的结果：

图 20

加载到主连合约后，模型的盈利能力大大降低，说明模型在多种行情中的适应力并不是很好。

除此之外，是否还有其他因素影响模型的盈利情况呢？

我们看一下主连合约的日 K 线图，图 21

图 21

在上面图中，我们可以看出：主连合约的每一次主力合约切换都不同程度的出现由于换月导

致的跳空现象，这些跳空并不是实际行情走势，在真实的交易中并不能带来真正的盈利或者

亏损，所以使用主连合约进行历史回测，得到的结果也会存在较大的误差！

我们怎么才能得到更真实的回测结果呢？

我们把模型加载到指数合约上，得到的测试结果如图 22

图 22

模型加载到指数合约上，同样的回测时间，盈利比加载到主连合约上盈利能力要明显提高，

主要的原因是指数合约更加连续，更能反映品种的趋势。

但指数合约是虚拟合约，并不能真正的进行交易，所以测试出的结果还不是真正的交易结果，

我们在模型中加入 TRADE_OTHER('AUTO');，实现在指数合约上计算模型信号，在具体月份

的主力合约上进行交易，并且主力合约切换的时候自动进行移仓，得到图 23 的测试结果

图 23

当模型中加入 TRADE_OTHER('AUTO')后，测试出的结果才是模型加载到指数上最后真实的交

易结果。

TRADE_OTHER 函数说明（详细说明请参考函数列表）：

TRADE_OTHER('CODE') 指定 CODE 合约为交易合约，CODE 为合约代码。

5、减少隔日跳空对短线交易的影响-TRADE_SMOOTHING

日内交易或者较短周期的波段交易，对于数据的连续性要求较高。但价格经常会由于某些原

因出现开盘跳空的情况，导致图表不连续，指标失真，给交易带来很大的影响，尤其是逆势

高开低走（低开高走）的行情，影响更大！

我们看一下图 24 和图 25，蓝色 k线是没有消除隔日跳空的行情，可以明显看出，消除隔日

跳空后，行情的连续性较高，对于行情的把握更加准确

图 24

图 25

我们如何将隔日跳空消除，并且使用消除隔日跳空后的数据进行模型的信号计算呢？

在模型中加入 TRADE_SMOOTHING 函数，就可以实现消除隔日跳空，函数说明如下（详细说

明请参考函数列表）：

TRADE_SMOOTHING 消除隔日跳空函数

用法：

模型中含有 TRADE_SMOOTHING 函数，主图 K 线为消除隔日跳空后计算得到的 K 线，模型根据消除跳空后 K

线取值计算

我们通过下面案例了解 TRADE_SMOOTHING 函数的编写方法及作用：

一个简单的唐迁通道模型，加载在白糖 1605 合约，模型代码如下，测试报告见图 26

HH:=HV(H,20);

LL:=LV(L,20);

CROSSUP(C,HH),BPK;

CROSSDOWN(C,LL),SPK;

AUTOFILTER;

图 26

加入 TRADE_SMOOTHING 函数后的模型，代码如下，测试报告见图 27

TRADE_SMOOTHING;

HH:=HV(H,20);

LL:=LV(L,20);

CROSSUP(C,HH),BPK;

CROSSDOWN(C,LL),SPK;

AUTOFILTER;

图 27

从测试报告中，我们可以看出，进行消除隔日跳空后的模型效果比之前回撤大幅降低，同时

在盈利率和胜率都有提高，模型综合表现有了明显提高。消除隔日跳空后的白糖 1605 的 K

线图见图 28，蓝色 k线是没有消除跳空的行情。

图 28

消除隔日跳空对于短线交易、日内交易策略来说并不是必须的，是否需要消除隔日跳空要视

具体策略而定。

6、从资金层面俯视来对信号执行做调控-IDLE

在研发模型的过程中，我们经常发现，模型在不同类型的行情中表现差异很大，甚至是在市

场发生了某种变化之后，模型彻底失效，但是我们又很难判定模型是失效还是正常的回撤，

导致出现了比较大的资金回撤，对整个组合甚至账户造成风险。

图 29

图 29 中的模型，在回测时间的前 1/3 的行情中，都是表现不错的，但是随着市场的变化，

模型在随后的行情中不在盈利，出现比较大的回撤，产生较大的风险，见图 30 的回测报告：

图 30

从回测报告中我们可以看出，模型在运行了一段时间以后出现了近 40%的回撤，这是我们在

实盘交易中很难接受的。

如果我们没有在历史回测中发现这种大的资金回撤，而是实盘中行情突然的变故导致大回

撤，后果就会比较严重。我们怎么减少这样的风险呢？

我们可以在模型中加入 IDLE 函数，实现不改变即有的交易信号，在资金出现较大的回撤的

时候停止交易，防止不可预知的风险扩大。

IDLE函数如下（更加具体函数说明请在函数列表中查阅）：

IDLE(COND) 限制开仓信号发出委托

用法：IDLE(COND)，当开仓信号发出时，如果 COND 条件成立，该信号不委托。IDLE 函数对平仓信号不起

作用，有持仓时即使满足 COND 也可以平仓。

我们通过这个案例看一下 IDLE 函数的用法和编写：

这是一个均线模型，在白糖指数上的测试效果，代码如下：

MA10:=MA(C,20);

MA30:=MA(C,30);

C>MA30,BK;

C<MA30,SK;

C>MA10,BP;

C<MA10,SP;

AUTOFILTER;

测试效果如上面的图 29 和图 30。

为了解决 40%的回撤问题，我们在模型中加入了 IDLE 函数，代码如下

MA10:=MA(C,20);

MA30:=MA(C,30);

GG:=HHV(MONEYTOT,0);

C>MA30,BK;

C<MA30,SK;

C>MA10,BP;

C<MA10,SP;

IDLE((MONEYTOT<GG*0.95&&MONEYTOT>GG*0.92)||MONEYTOT<GG*0.85);//权益回撤一定幅度

停止交易

AUTOFILTER;

加入 IDLE 函数后，在设定的回撤阀值内，即使模型出现信号，但不下单，很好的避免较大

的回撤，具体效果如图 31 和图 32

图 31

图 32

从图 31 中，我们可以看到两条资金曲线，黄色的资金线是加入 IDLE 之后的效果，很明显，

资金在出现大回撤的之后没有产生交易，风险要小很多。从图 32 中的详细测试报告我们也

可以看到，模型的最大回撤降低到 22%，同时平均盈亏、胜率等方面都有很好的提升。

